```
> restart:with(plots):
```

monkey.mws

The Monkey Saddle as a parametric Monge surface.

First, a picture:

```
> R:=[u,v,u^3-3*u*v^2];
>
```

$$R := [u, v, u^3 - 3\,u\,v^2]$$

```
> plot3d(R(u,v),u=-Pi..Pi,v=-Pi..Pi,shading=zgreyscale,lightmodel=light4,axes=framed,style=PATCHNOGRID);
```



Now restart the computation of the gauss and mean curvature using Cartan's Repere Mobile

```
> restart:with(linalg):
Warning, new definition for norm
Warning, new definition for trace
```

```
> with(difforms):
> with(liesymm): setup(u,v):
Warning, new definition for `&^`
Warning, new definition for close
Warning, new definition for d
Warning, new definition for mixpar
Warning, new definition for wdegree
```

The Position Vector in R3 parametrized with (u,v). The example is for a Monge Surface z=g(u,v)

```
> RR:=[u,v,u^3-3*u*v^2];
```

$$RR := [u, v, u^3 - 3\,u\,v^2]$$

```
> XX:=RR[1]:YY:=RR[2]:ZZ:=RR[3]:
> Yu:=diff(RR,u):
```

```
> Yv:=diff(RR,v):
> NNU:=crossprod(Yu,Yv):
```
Scale the adjoint normal field here by rho
```
> rho:=innerprod(NNU,NNU)^(1/2):
> #rho:=1:
>
```
This vector (surface normal) NNU can be computed from the Adjoint Matrix operation on the two tangent vectors Yu and Yv. The basis frame utilizes this surface normal with arbitrary scaling
```
> NN:=([factor(NNU[1]),factor(NNU[2]),simplify(factor(NNU[3]))]):
> FF:=array([[Yu[1],Yv[1],NN[1]/rho],[Yu[2],Yv[2],NN[2]/rho],[Yu[3],Yv[3],NN[3]/rh
o]  ] );
```
$$FF := \begin{bmatrix} 1 & 0 & -3\frac{(u-v)(u+v)}{\sqrt{9u^4+18u^2v^2+9v^4+1}} \\ 0 & 1 & 6\frac{u\,v}{\sqrt{9u^4+18u^2v^2+9v^4+1}} \\ 3u^2-3v^2 & -6\,u\,v & \frac{1}{\sqrt{9u^4+18u^2v^2+9v^4+1}} \end{bmatrix}$$

The Repere Mobile or FRAME MATRIX, FF. note that the frame matrix is not orthonormal.!!
```
> detFF:=simplify((det(FF)));
```
$$detFF := \sqrt{9u^4+18u^2v^2+9v^4+1}$$

For the Monge casel the deteminant is non-zero globally, hence an inverse always exists.
```
> FFINVD:=evalm(FF^(-1)):
```
The 1-form components of the differential position vector with respect to the Basis Frame, F.
```
> dR:=innerprod(FFINVD,[d(XX),d(YY),d(ZZ)]):
> sigma1:=(wcollect(dR[1])):
> sigma2:=wcollect(dR[2]):
```
Note that sigma1 is du and sigma2 is dv for a parametric Monge surfaces!!
```
> omega:=(wcollect(dR[3]));
```
$$\omega := 0$$

Note that this term vanishes for a parametric Monge surface, hence parametric Monge surfaces exhibit no TORSION!! of the Affine type (that is there is no translational shear defects!)
```
>
>
```
Compute the Cartan Matrix of connection forms from C=[F(inverse)] times d[F]
```
> dFF:=array([[d(FF[1,1]),d(FF[1,2]),d(FF[1,3])],[d(FF[2,1]),d(FF[2,2]),d(FF[2,3])
],[d(FF[3,1]),d(FF[3,2]),d(FF[3,3])]]):
> cartan:=evalm(FFINVD&*dFF):
```
The interior connection coefficients (can be Christoffel symbols on the parameter space
```
> Gamma11:=(wcollect(cartan[1,1])):
> Gamma12:=(wcollect(cartan[1,2])):
> Gamma21:=(wcollect(cartan[2,1])):
> Gamma22:=(wcollect(cartan[2,2])):
```
The second fundamental form or shape matrix comes from the third row of the Cartan matrix
```
> h1:=wcollect(cartan[3,1]):
> gamma1:=wcollect(cartan[1,3]):
> h2:=(wcollect(cartan[3,2])):
> gamma2:=(wcollect(cartan[2,3])):
```

The abnormality for the parametric surface will show up as a non-zero entry in the [3,3] slot of the Cartan Matrix. Always an exact differential for parametric and Monge surfaces. Therefore implicit Monge surfaces will admit disclination defects (Torsion of the second kind due to rotations)

```
> Omega:=(wcollect(factor(simpform(cartan[3,3])))):
```

Omega vanishes for a given normalization.

```
> wcollect(factor(simpform(d(Omega)))):
> FROBOMEGA:=simpform(Omega&^d(Omega)):
```

The coefficients of the shape matrix determined from the Cartan matrix.

```
> factor(simpform(Omega&^gamma1)):
> simplify(Omega&^gamma1):
```

The components of the disclination 2-form are given above. Note that they are proportional to the Square Root of the Gauss Curvature (for scaling = 1) and form the "Stream" vector relative to the gradient of the Monge function g -- a symplectic rotation

```
> shape11:=-factor(gamma1&^d(v)/d(u)&^d(v)):
> shape12:=-factor(gamma1&^d(u)/d(v)&^d(u)):
> shape21:=-factor(gamma2&^d(v)/d(u)&^d(v)):
> shape22:=-factor(gamma2&^d(u)/d(v)&^d(u)):
>
> SHAPE:=array([[shape11,shape12],[shape21,shape22]]):
> HH:=simplify(trace(SHAPE)/2):
> print(`Mean Curvature is `,HH);
```

$$\textit{Mean Curvature is },\ -27\,\frac{u\,(-2\,u^2\,v^2-3\,v^4+u^4)}{(9\,u^4+18\,u^2\,v^2+9\,v^4+1)^{3/2}}$$

```
> KK:=simplify(det(SHAPE)):
> print(`Gauss Curvature is `,KK);
```

$$\textit{Gauss Curvature is },\ -36\,\frac{u^2+v^2}{(9\,u^4+18\,u^2\,v^2+9\,v^4+1)^{2}}$$

```
>
```

Note that the scaling of the normal or adjoint vector is a common factor of the formulas for the mean curvature and the Gauss curvature. Note the appearance of the Hessian of the Monge function.
The induce metric appears below

```
> GUN:=innerprod(transpose(FF),FF);
```

$$GUN := \begin{bmatrix} 1+9\,u^4-18\,u^2\,v^2+9\,v^4 & -18\,(u^2-v^2)\,u\,v & 0 \\ -18\,(u^2-v^2)\,u\,v & 1+36\,u^2\,v^2 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
>
>
>
```